1. <start> -> require "ifj21"<program>

2. <program> -> <fnc_dec> <program>
3. <program> -> <fnc_call> <program>
4. <program> -> <fnc_def> <program>
5. <program> -> ε

6. <fnc_dec> -> global id_fnc : function(<params_dec>) <return_type>
7. <params_dec> -> ε
8. <params_dec> -> <data_type> <params_dec2>
9. <params_dec2> -> ε
10. <params_dec2> -> ,<data_type><params_dec2>
11. <return_type> -> : <data_type>
12. <return_type> -> ε
13. <data_type> -> integer
14. <data_type> -> number
15. <data_type> -> string

16. <fnc_call> -> id_fnc (<value>)
17. <value> -> ε
18. <value> -> <value_last> <value2>
19. <value2> -> ,<value_last> <value2>
20. <value2> -> ε
21. <value_last> -> id_var
22. <value_last> -> integer_value
23. <value_last> -> number_value
24. <value_last> -> string_value

25. <fnc_def> -> <fnc_head> <fnc_def2> end
26. <fnc_head> -> function id_fnc (<params_def>)
27. <fnc_def2> -> <fnc_body> <return_void>
28. <fnc_def2> -> : <data_type><fnc_body><return>
29. <params_def> -> ε
30. <params_def> -> <var_def> <params_def2>
31. <params_def2> -> ε
32. <params_def2> -> ,<var_def> <params_def2>
33. <var_def> -> id_var : <data_type>

34. <return> -> return <expr>

35. <return_void> -> return
36. <return_void> -> ε

37. <fnc_body> -> <if><fnc_body>
38. <fnc_body> -> <loop><fnc_body>
39. <fnc_body> -> <statement><fnc_body>
40. <fnc_body> -> ε

41. <statement> -> <var_dec>
42. <statement> -> id_var = <var_assign>
43. <statement> -> <fnc_call>

44. <var_dec> -> local <var_def> <var_dec_init>
45. <var_dec_init> -> = <var_dec_init2>
46. <var_dec_init> -> ε
47. <var_dec_init2> -> <expr>
48. <var_dec_init2> -> <fnc_call>

49. <var_assign> -> <expr>
50. <var_assign> -> <fnc_call>

51. <if> -> if <expr> then <statements> else <statements> end
52. <loop> -> while <expr> do <statements> end

53. <statements> -> ε
54. <statements> -> <statement> <statements>

POZN.: Slova a znaky psaná černě jsou neterminály.

POZN.2: Znak mezery v pravidlech je irelevantní a nic neznačí, slouží pouze k zpřehlednění.

POZN.3: Neterminál <expr> není rozšířen a zpracuje ho analýza shora dolů.